;;;;;;;;;;;;;;; CONSTANTS ;;;;;;;;;;;;;;;

globals [
  maxspec  ; the upper bound of the numerical values for the receptors, epitoptes, paratopes, etc, used in the simulation

  numAbs   ; the number of antibodies secreted by plasma cells at each time step
  numCyto  ; the number of cytokines produced by effector T helper cells at each time step

  life_B      ; half-life of a B cell
  life_Plasma ; half-life of a plasma cell 
  life_MemB   ; the half-life of a memory B cell
  life_Th     ; the half-life of a naive T helper cell
  life_MemTh  ; the half-life of a memory T helper cell
  life_EffTh  ; the half-life of an effector T helper cell
  life_Tc     ; the half-life of a naive T cytotoxic (Tc) cell
  life_MemTc  ; the half-life of a memory Tc cell
  life_EffTc  ; the half-life of an effector Tc cell
  life_Ab     ; the half-life of an antibody
  life_Cyto   ; the half-life of a cytokine
  life_Apc    ; the half-life of an antigen-presenting cell (apc)
  life_Virion ; the half-life of a virion - an extracellular viral particle

  pdissociation  ; the probability that an APC or, in this case, a follicular denritic cell bound to hiv will dissociate at a given time step
  p_nonspec      ; probability of successful binding in a nonspecific interaction, e.g., APC/Ag

  bystander_thr

  ; flags
  pmutB  ; the B cell hypermutation rate
  pmutV  ; the viral mutation rate

  ; statistics
  neutralized  ; keeps track of the total number of virions neutralized by antibodies
  killed      ;  keeps track of the total number of infected cells killed by effector Tc cells
]

to initialize-globals
  set maxspec 3

  set numAbs 2
  set numCyto 2

  set life_B 15
  set life_Plasma 5
  set life_MemB 10
  set life_Th 15
  set life_MemTh 10
  set life_EffTh 5
  set life_Tc 15
  set life_MemTc 10
  set life_EffTc 5
  set life_Ab 2
  set life_Cyto 2
  set life_Apc 15
  set life_Virion 1000

  set pdissociation 0.001
  set p_nonspec 0.5

  set bystander_thr 5

  set pmutV 0
  set pmutB 0

  set neutralized 0
  set killed 0
end










;;;;;;;;;;;;;;; STRUCTURE ;;;;;;;;;;;;;;;

patches-own [  ; used for the diffusion of cells
  cytos  ; the number of cytokines on a given patch - used to simulating chemotaxis
  inf    ; the number of infected cells on a given patch - used to simulate chemotaxis
]

breed [ hiv ]
breed [ apc ]
breed [ bCell ]
breed [ memB ]
breed [ plasma ]
breed [ ab ]
breed [ thCell ]
breed [ memTh ]
breed [ effTh ]
breed [ cyto ]
breed [ tcCell ]
breed [ memTc ]
breed [ effTc ]

turtles-own [  ; base class
  ; state depends on cell; may be: 'resting','activated','stimulated','infected'
  state

  ; steps since it was created
  steps
  halfLife  ; half-life: used for replenishment

  peptide  ; the numeric value of this cell's receptor
  ; compounds: paratope (antibody), epitope (antigen)

  viralLoad  ; viral load
  sinceInf   ; the number of time steps since the given turtle has been infected. The value of this variable, 
             ; together with the viral replication rate, is used to calculate the number of viral particles present in a 
             ; productively infected cell at any given time, post infection
]
apc-own [
  vRNA
]











;;;;;;;;;;;;;;; SETUP ;;;;;;;;;;;;;;;

to setup
  clear-all
  set-default-shape turtles "circle"
  random-seed 250

  initialize-globals
  create-cells
  ask patches [
    set inf 0
  ]
  reset-ticks
end

to initialize-turtle
  set state "resting"
  set steps 0
  set halfLife 1000
  set peptide random maxspec
  set sinceInf 0
  if breed = hiv [
    set color violet
    set shape "face sad"
  ]
  if breed = bCell [
    set halfLife life_B 
    set color green
  ]
  if breed = memB [
    set halfLife life_MemB 
    set color green
  ]
  if breed = plasma [
    set halfLife life_Plasma
    set color green
  ]
  if breed = ab [
    set halfLife life_Ab 
    set color yellow
  ]
  if breed = thCell [
    set halfLife life_Th 
    set color blue 
  ]
  if breed = memTh [
    set halfLife life_MemTh 
    set color blue 
  ]
  if breed = effTh [
    set halfLife life_EffTh
    set color blue
  ]
  if breed = cyto [
    set halfLife life_Cyto
    set color orange
  ]
  if breed = tcCell [
    set halfLife life_Tc 
    set color brown 
  ]
  if breed = memTc [
    set halfLife life_MemTc 
    set color brown
  ]
  if breed = effTc [
    set halfLife life_EffTc 
    set color brown
  ]
  if breed = apc [
    set halfLife life_Apc 
    set vRNA []
    set color grey
  ]
end

to create-cells
  create-bCell numB [
    initialize-turtle
    disperse
  ]
  create-hiv numHiv [
    initialize-turtle
    disperse
  ]
  create-thCell numTh [
    initialize-turtle
    disperse
  ]
  create-tcCell numTc [
    initialize-turtle
    disperse
  ]
  create-apc numApc [
    initialize-turtle
    disperse
  ]
end

to disperse
  move-to one-of patches
end





















;;;;;;;;;;;;;;; UPDATE ;;;;;;;;;;;;;;;

to update
  interact-all
  viral-production
  apoptosis
  cell-production
  apc-dissociation
  resupply
  do-diffuse
  tick
  ask turtles [set steps steps + 1]
end

to interact-hiv [_hiv]
  if breed = bCell or breed = memB [
    if state = "resting" and success _hiv self [
      set state "activated"
      set peptide [peptide] of _hiv
      ask _hiv [die]
    ]
  ]
  if breed = apc [  ; apcs (particularly folicular dendritic cells) may serve as viral reservoirs
    if probability p_nonspec [
      set state "activated"
      set peptide [peptide] of _hiv  ; an activated apc presents viral peptides on both HLA1 and HLA2
      set vRNA lput [peptide] of _hiv vRNA
      ask _hiv [die]
    ]
  ]
  ; the neutralization of a virion by an antibody. Note that to keep the number of turtles relatively low and, consequently, improve the
  ; speed of the simulation, all neutralized virions and the antibodies involved are immediately removed from the simulation.
  if breed = ab [
    if success _hiv self and probability neutRate [
      set neutralized neutralized + 1
      ask _hiv [die]
    ]
  ]
  ; viral infection of target cells. The probability of infection is determined by the infectivity parameter or the lethality variable
  if breed != hiv and breed != cyto and breed != ab [
    if probability infectivity [
      set state "infected"
      set peptide [peptide] of _hiv
      ask _hiv [die]
    ]
  ]
end

to interact-apc [_apc]  ; please notice apc must be 'activated'
  if (breed = thCell or breed = memTh or breed = tcCell or breed = memTc) and (success _apc self) [
    set state "stimulated"
  ]
end

; the interactions of entities, as described in the supplemental materials
to interact-all
  ; interactions involving hiv
  ask hiv [
    let _hiv self
    ask turtles-here [
      interact-hiv _hiv
    ]
  ]
  ; interactions involving apc
  ask apc with [state = "activated"] [
    let _apc self
    ask turtles-here [
      interact-apc _apc
    ]
  ]
  ; customized interactions
  ask turtles [
    if breed = thCell or breed = memTh [
      let _thcell self
      ask turtles-here [
        ; B cells are also APCs and can thus stimulate Th cells
        if (breed = bCell or breed = memB) and (state = "activated") and (success self _thcell) [
          set state "stimulated"  ; FIXME: does it make sense for B to become stimulated?
          ask _thcell [set state "stimulated"]
        ]
      ]
    ]
  ]
  ; the killing of infected cells by Tc cells. The probability that a Tc cell will kill an infected cell on initial contact is given
  ; by the ctlRate parameter
  ask effTc [
    let this self
    ask turtles-here with [state = "infected"] [
      if (success peptide [peptide] of this) and (probability ctlRate) [
        set killed killed + 1
        die
      ]
    ]
  ]
  ; cells finding a lot of cytokines
  ask turtles [
    if breed = bCell or breed = memB or breed = tcCell or breed = memTc [
      if (count cyto-here) >= bystander_thr [
        set state "stimulated"
      ]
    ]
  ]
end

; productively infected cells (as opposed to infected but dormant/anergic cells)  continue to produce new virions until
; either a) they are killed during CTL or (b) their viral load exceeds the lethal load of the infecting virus. In the latter
; case, the infected cells undergo lysis, releasing new virions into the simulation.
to viral-production
  ask turtles with [state = "infected"] [
    let this self
    set viralLoad floor(exp(0.05 * sinceInf))
    ifelse viralLoad > 24 [
      hatch viralLoad [
        set breed hiv
        initialize-turtle
        ifelse probability pmutV [
          set peptide random maxspec 
        ][
        set peptide [peptide] of this
        ]
      ]
      die
    ][
    set sinceInf sinceInf + 1
    ]
  ]
end

; the half-life of each entity is used to determine the probability that it will die during a given time step. Because the gradual 
; depletion of Th cells, a hallmark of hiv pathogenesis, is thought to arise not only from direct hiv-mediated killing of infected 
; Th cells but also from indirect killing (or bystander effect) of Th cells. We assume 
; that this indirect killing of Th cells is a function of the number of both activated and infected cells in the vicinity of a 
; Th cell and on the value of the AICD parameter.
to apoptosis
  ask turtles [
    if breed = thCell or breed = memTh or breed = effTh [
      set halfLife max list 0 (halfLife - (aicd * ((count turtles-here with [state = "infected"]) + (count hiv-here))))
    ]
    let diff (ln 2) * halfLife
    if probability exp (- diff) [
      die
    ]
  ]
end

; 2 memory cells
to cell-production
  ; plasma cells secrete numAbs antibodies during each time step
  ask ab [
    let this self
    hatch numAbs [
      set breed ab
      initialize-turtle
      set peptide [peptide] of this
    ]
  ]
  ; effector Th cells secrete numCyto cytokines during each time step
  ask effTh [
    hatch numCyto [
      set breed cyto
      initialize-turtle
    ]
  ]    
  ; activated B, memory B, Th, memory Th, Tc, and memory Tc cells divide over two time steps to produce 4 cells - 2 effectors cells and
  ask turtles [
    let this self
    if steps >= 2 and state = "stimulated" [  ; cell division
      if breed = bCell or breed = memB [
        hatch 2 [
          set breed memB
          initialize-turtle
          ifelse(probability pmutB) [
            set peptide random maxspec
          ][
          set peptide [peptide] of this
          ]
        ]
        hatch 2 [
          set breed plasma
          initialize-turtle
          set peptide [peptide] of this
        ]
        die
      ]
      if breed = thCell or breed = memTh [
        hatch 2 [
          set breed memTh
          initialize-turtle
          set peptide [peptide] of this
        ]
        hatch 2 [
          set breed effTh
          initialize-turtle
          set peptide [peptide] of this
        ]
        die
      ]
      if breed = tcCell or breed = memTc [
        hatch 2 [
          set breed memTc
          initialize-turtle
          set peptide [peptide] of this
        ]
        hatch 2 [
          set breed effTc
          initialize-turtle
          set peptide [peptide] of this
        ]
        die
      ]
    ]
  ]
end

; the dissociation of an apc from a bound virion. This is thought to contribute to the pepertuation of viremia in chronically 
; infected patients.
to apc-dissociation
  ask apc [
    let this self
    if probability pdissociation and not empty? vRNA [
      foreach vRNA [
        hatch 1 [
          set breed hiv
          initialize-turtle
          set peptide ?
        ]
      ]
      set vRNA []
      set state "resting"
    ]
  ]
end

; to maintain a relatively constant supply of naive lymphocytes, these cells are replenished by the bone marrow and thymus at the same rate
; as they undergo apoptotic death  
to resupply
  repeat numTh [
    if probability exp(- ((ln 2) * life_Th)) [
      create-thCell 1 [
        initialize-turtle
      ]
    ]
  ]
  repeat numTc [
    if probability exp(- ((ln 2) * life_Tc)) [
      create-tcCell 1 [
        initialize-turtle
      ]
    ]
  ]
  repeat numB [
    if probability exp(- ((ln 2) * life_B)) [
      create-bCell 1 [
        initialize-turtle
      ]
    ]
  ]
  repeat numApc [
    if probability exp(- ((ln 2) * life_Apc)) [
      create-apc 1 [
        initialize-turtle
      ]
    ]
  ]
end

; the difussion of entities is assumed to be driven by a chemotactic gradient. For example, naive Tc cells diffuse in the 
; direction of greatest cytokine concentration, while effector Tc cells seek out sites with high numbers of infected cells.
to do-diffuse
  ask patches  [ 
    set inf count turtles-here with [state = "infected"]
    set cytos count cyto-here
  ]
  ask turtles [
    repeat 2 [
      ifelse breed = tcCell or breed = memTc or breed = bCell or breed = memB [
        ;uphill cytos
        move-to max-one-of neighbors [cytos]
      ][
      ifelse breed = effTc [ 
        ;uphill inf
        move-to max-one-of neighbors [inf]
      ][
      move-to one-of neighbors  ; random walk
      ]
      ]
    ]
  ]
end








;;;;;;;;;;;;;;; UTILITIES ;;;;;;;;;;;;;;;

; this procedure determines the probability of successful interaction (e.g., bond formation) between two entities
to-report success [agent1 agent2]
  let receptor1 [peptide] of agent1
  let receptor2 [peptide] of agent2
  let var abs(receptor1 - receptor2)
  report probability exp(- var)
end

; this procedure determines the outcome of events (i.e. true or false) based on the probabilities of the events
to-report probability [p]
  report random-float 1 < p
end
@#$#@#$#@
GRAPHICS-WINDOW
230
30
650
471
20
20
10.0
1
10
1
1
1
0
1
1
1
-20
20
-20
20
1
1
1
ticks
30.0

BUTTON
20
30
105
63
NIL
setup
NIL
1
T
OBSERVER
NIL
NIL
NIL
NIL
1

BUTTON
115
30
207
63
go
update
T
1
T
OBSERVER
NIL
NIL
NIL
NIL
1

SLIDER
20
105
112
138
numB
numB
0
100
50
1
1
NIL
HORIZONTAL

SLIDER
20
240
145
273
infectivity
infectivity
0
1
0.007
0.0010
1
NIL
HORIZONTAL

SLIDER
20
175
112
208
numHiv
numHiv
0
300
10
1
1
NIL
HORIZONTAL

SLIDER
115
140
207
173
numTh
numTh
0
100
50
1
1
NIL
HORIZONTAL

SLIDER
115
105
207
138
numApc
numApc
0
100
55
1
1
NIL
HORIZONTAL

SLIDER
20
140
112
173
numTc
numTc
0
100
50
1
1
NIL
HORIZONTAL

SLIDER
20
275
145
308
neutRate
neutRate
0
1
0.109
0.0010
1
NIL
HORIZONTAL

SLIDER
20
310
145
343
ctlRate
ctlRate
0
100
0.85
0.0010
1
NIL
HORIZONTAL

SLIDER
20
345
145
378
AICD
AICD
0
100
0.11
0.01
1
NIL
HORIZONTAL

PLOT
665
30
980
215
B family
time
count
0.0
10.0
0.0
10.0
true
true
"" ""
PENS
"B" 1.0 0 -10899396 true "" "plot count bCell + count memB"
"ab" 1.0 0 -1184463 true "" "plot count ab"

PLOT
665
225
980
405
Th and HIV
time
count
0.0
10.0
0.0
10.0
true
true
"" ""
PENS
"Th" 1.0 0 -13345367 true "" "plot count thCell + count memTh"
"hiv" 1.0 0 -8630108 true "" "plot count hiv"

PLOT
665
415
980
600
Th family
time
count
0.0
10.0
0.0
10.0
true
true
"" ""
PENS
"Th" 1.0 0 -13345367 true "" "plot count thCell"
"MemTh" 1.0 0 -8630108 true "" "plot count memTh"
"EffTh" 1.0 0 -2674135 true "" "plot count effTh"
"Cyto" 1.0 0 -955883 true "" "plot count cyto"

@#$#@#$#@
## WHAT IS IT?

(a general understanding of what the model is trying to show or explain)

## HOW IT WORKS

(what rules the agents use to create the overall behavior of the model)

## HOW TO USE IT

(how to use the model, including a description of each of the items in the Interface tab)

## THINGS TO NOTICE

(suggested things for the user to notice while running the model)

## THINGS TO TRY

(suggested things for the user to try to do (move sliders, switches, etc.) with the model)

## EXTENDING THE MODEL

(suggested things to add or change in the Code tab to make the model more complicated, detailed, accurate, etc.)

## NETLOGO FEATURES

(interesting or unusual features of NetLogo that the model uses, particularly in the Code tab; or where workarounds were needed for missing features)

## RELATED MODELS

(models in the NetLogo Models Library and elsewhere which are of related interest)

## CREDITS AND REFERENCES

(a reference to the model's URL on the web if it has one, as well as any other necessary credits, citations, and links)
@#$#@#$#@
default
true
0
Polygon -7500403 true true 150 5 40 250 150 205 260 250

airplane
true
0
Polygon -7500403 true true 150 0 135 15 120 60 120 105 15 165 15 195 120 180 135 240 105 270 120 285 150 270 180 285 210 270 165 240 180 180 285 195 285 165 180 105 180 60 165 15

arrow
true
0
Polygon -7500403 true true 150 0 0 150 105 150 105 293 195 293 195 150 300 150

box
false
0
Polygon -7500403 true true 150 285 285 225 285 75 150 135
Polygon -7500403 true true 150 135 15 75 150 15 285 75
Polygon -7500403 true true 15 75 15 225 150 285 150 135
Line -16777216 false 150 285 150 135
Line -16777216 false 150 135 15 75
Line -16777216 false 150 135 285 75

bug
true
0
Circle -7500403 true true 96 182 108
Circle -7500403 true true 110 127 80
Circle -7500403 true true 110 75 80
Line -7500403 true 150 100 80 30
Line -7500403 true 150 100 220 30

butterfly
true
0
Polygon -7500403 true true 150 165 209 199 225 225 225 255 195 270 165 255 150 240
Polygon -7500403 true true 150 165 89 198 75 225 75 255 105 270 135 255 150 240
Polygon -7500403 true true 139 148 100 105 55 90 25 90 10 105 10 135 25 180 40 195 85 194 139 163
Polygon -7500403 true true 162 150 200 105 245 90 275 90 290 105 290 135 275 180 260 195 215 195 162 165
Polygon -16777216 true false 150 255 135 225 120 150 135 120 150 105 165 120 180 150 165 225
Circle -16777216 true false 135 90 30
Line -16777216 false 150 105 195 60
Line -16777216 false 150 105 105 60

car
false
0
Polygon -7500403 true true 300 180 279 164 261 144 240 135 226 132 213 106 203 84 185 63 159 50 135 50 75 60 0 150 0 165 0 225 300 225 300 180
Circle -16777216 true false 180 180 90
Circle -16777216 true false 30 180 90
Polygon -16777216 true false 162 80 132 78 134 135 209 135 194 105 189 96 180 89
Circle -7500403 true true 47 195 58
Circle -7500403 true true 195 195 58

circle
false
0
Circle -7500403 true true 0 0 300

circle 2
false
0
Circle -7500403 true true 0 0 300
Circle -16777216 true false 30 30 240

cow
false
0
Polygon -7500403 true true 200 193 197 249 179 249 177 196 166 187 140 189 93 191 78 179 72 211 49 209 48 181 37 149 25 120 25 89 45 72 103 84 179 75 198 76 252 64 272 81 293 103 285 121 255 121 242 118 224 167
Polygon -7500403 true true 73 210 86 251 62 249 48 208
Polygon -7500403 true true 25 114 16 195 9 204 23 213 25 200 39 123

cylinder
false
0
Circle -7500403 true true 0 0 300

dot
false
0
Circle -7500403 true true 90 90 120

face happy
false
0
Circle -7500403 true true 8 8 285
Circle -16777216 true false 60 75 60
Circle -16777216 true false 180 75 60
Polygon -16777216 true false 150 255 90 239 62 213 47 191 67 179 90 203 109 218 150 225 192 218 210 203 227 181 251 194 236 217 212 240

face neutral
false
0
Circle -7500403 true true 8 7 285
Circle -16777216 true false 60 75 60
Circle -16777216 true false 180 75 60
Rectangle -16777216 true false 60 195 240 225

face sad
false
0
Circle -7500403 true true 8 8 285
Circle -16777216 true false 60 75 60
Circle -16777216 true false 180 75 60
Polygon -16777216 true false 150 168 90 184 62 210 47 232 67 244 90 220 109 205 150 198 192 205 210 220 227 242 251 229 236 206 212 183

fish
false
0
Polygon -1 true false 44 131 21 87 15 86 0 120 15 150 0 180 13 214 20 212 45 166
Polygon -1 true false 135 195 119 235 95 218 76 210 46 204 60 165
Polygon -1 true false 75 45 83 77 71 103 86 114 166 78 135 60
Polygon -7500403 true true 30 136 151 77 226 81 280 119 292 146 292 160 287 170 270 195 195 210 151 212 30 166
Circle -16777216 true false 215 106 30

flag
false
0
Rectangle -7500403 true true 60 15 75 300
Polygon -7500403 true true 90 150 270 90 90 30
Line -7500403 true 75 135 90 135
Line -7500403 true 75 45 90 45

flower
false
0
Polygon -10899396 true false 135 120 165 165 180 210 180 240 150 300 165 300 195 240 195 195 165 135
Circle -7500403 true true 85 132 38
Circle -7500403 true true 130 147 38
Circle -7500403 true true 192 85 38
Circle -7500403 true true 85 40 38
Circle -7500403 true true 177 40 38
Circle -7500403 true true 177 132 38
Circle -7500403 true true 70 85 38
Circle -7500403 true true 130 25 38
Circle -7500403 true true 96 51 108
Circle -16777216 true false 113 68 74
Polygon -10899396 true false 189 233 219 188 249 173 279 188 234 218
Polygon -10899396 true false 180 255 150 210 105 210 75 240 135 240

house
false
0
Rectangle -7500403 true true 45 120 255 285
Rectangle -16777216 true false 120 210 180 285
Polygon -7500403 true true 15 120 150 15 285 120
Line -16777216 false 30 120 270 120

leaf
false
0
Polygon -7500403 true true 150 210 135 195 120 210 60 210 30 195 60 180 60 165 15 135 30 120 15 105 40 104 45 90 60 90 90 105 105 120 120 120 105 60 120 60 135 30 150 15 165 30 180 60 195 60 180 120 195 120 210 105 240 90 255 90 263 104 285 105 270 120 285 135 240 165 240 180 270 195 240 210 180 210 165 195
Polygon -7500403 true true 135 195 135 240 120 255 105 255 105 285 135 285 165 240 165 195

line
true
0
Line -7500403 true 150 0 150 300

line half
true
0
Line -7500403 true 150 0 150 150

pentagon
false
0
Polygon -7500403 true true 150 15 15 120 60 285 240 285 285 120

person
false
0
Circle -7500403 true true 110 5 80
Polygon -7500403 true true 105 90 120 195 90 285 105 300 135 300 150 225 165 300 195 300 210 285 180 195 195 90
Rectangle -7500403 true true 127 79 172 94
Polygon -7500403 true true 195 90 240 150 225 180 165 105
Polygon -7500403 true true 105 90 60 150 75 180 135 105

plant
false
0
Rectangle -7500403 true true 135 90 165 300
Polygon -7500403 true true 135 255 90 210 45 195 75 255 135 285
Polygon -7500403 true true 165 255 210 210 255 195 225 255 165 285
Polygon -7500403 true true 135 180 90 135 45 120 75 180 135 210
Polygon -7500403 true true 165 180 165 210 225 180 255 120 210 135
Polygon -7500403 true true 135 105 90 60 45 45 75 105 135 135
Polygon -7500403 true true 165 105 165 135 225 105 255 45 210 60
Polygon -7500403 true true 135 90 120 45 150 15 180 45 165 90

sheep
false
15
Circle -1 true true 203 65 88
Circle -1 true true 70 65 162
Circle -1 true true 150 105 120
Polygon -7500403 true false 218 120 240 165 255 165 278 120
Circle -7500403 true false 214 72 67
Rectangle -1 true true 164 223 179 298
Polygon -1 true true 45 285 30 285 30 240 15 195 45 210
Circle -1 true true 3 83 150
Rectangle -1 true true 65 221 80 296
Polygon -1 true true 195 285 210 285 210 240 240 210 195 210
Polygon -7500403 true false 276 85 285 105 302 99 294 83
Polygon -7500403 true false 219 85 210 105 193 99 201 83

square
false
0
Rectangle -7500403 true true 30 30 270 270

square 2
false
0
Rectangle -7500403 true true 30 30 270 270
Rectangle -16777216 true false 60 60 240 240

star
false
0
Polygon -7500403 true true 151 1 185 108 298 108 207 175 242 282 151 216 59 282 94 175 3 108 116 108

target
false
0
Circle -7500403 true true 0 0 300
Circle -16777216 true false 30 30 240
Circle -7500403 true true 60 60 180
Circle -16777216 true false 90 90 120
Circle -7500403 true true 120 120 60

tree
false
0
Circle -7500403 true true 118 3 94
Rectangle -6459832 true false 120 195 180 300
Circle -7500403 true true 65 21 108
Circle -7500403 true true 116 41 127
Circle -7500403 true true 45 90 120
Circle -7500403 true true 104 74 152

triangle
false
0
Polygon -7500403 true true 150 30 15 255 285 255

triangle 2
false
0
Polygon -7500403 true true 150 30 15 255 285 255
Polygon -16777216 true false 151 99 225 223 75 224

truck
false
0
Rectangle -7500403 true true 4 45 195 187
Polygon -7500403 true true 296 193 296 150 259 134 244 104 208 104 207 194
Rectangle -1 true false 195 60 195 105
Polygon -16777216 true false 238 112 252 141 219 141 218 112
Circle -16777216 true false 234 174 42
Rectangle -7500403 true true 181 185 214 194
Circle -16777216 true false 144 174 42
Circle -16777216 true false 24 174 42
Circle -7500403 false true 24 174 42
Circle -7500403 false true 144 174 42
Circle -7500403 false true 234 174 42

turtle
true
0
Polygon -10899396 true false 215 204 240 233 246 254 228 266 215 252 193 210
Polygon -10899396 true false 195 90 225 75 245 75 260 89 269 108 261 124 240 105 225 105 210 105
Polygon -10899396 true false 105 90 75 75 55 75 40 89 31 108 39 124 60 105 75 105 90 105
Polygon -10899396 true false 132 85 134 64 107 51 108 17 150 2 192 18 192 52 169 65 172 87
Polygon -10899396 true false 85 204 60 233 54 254 72 266 85 252 107 210
Polygon -7500403 true true 119 75 179 75 209 101 224 135 220 225 175 261 128 261 81 224 74 135 88 99

wheel
false
0
Circle -7500403 true true 3 3 294
Circle -16777216 true false 30 30 240
Line -7500403 true 150 285 150 15
Line -7500403 true 15 150 285 150
Circle -7500403 true true 120 120 60
Line -7500403 true 216 40 79 269
Line -7500403 true 40 84 269 221
Line -7500403 true 40 216 269 79
Line -7500403 true 84 40 221 269

wolf
false
0
Polygon -16777216 true false 253 133 245 131 245 133
Polygon -7500403 true true 2 194 13 197 30 191 38 193 38 205 20 226 20 257 27 265 38 266 40 260 31 253 31 230 60 206 68 198 75 209 66 228 65 243 82 261 84 268 100 267 103 261 77 239 79 231 100 207 98 196 119 201 143 202 160 195 166 210 172 213 173 238 167 251 160 248 154 265 169 264 178 247 186 240 198 260 200 271 217 271 219 262 207 258 195 230 192 198 210 184 227 164 242 144 259 145 284 151 277 141 293 140 299 134 297 127 273 119 270 105
Polygon -7500403 true true -1 195 14 180 36 166 40 153 53 140 82 131 134 133 159 126 188 115 227 108 236 102 238 98 268 86 269 92 281 87 269 103 269 113

x
false
0
Polygon -7500403 true true 270 75 225 30 30 225 75 270
Polygon -7500403 true true 30 75 75 30 270 225 225 270

@#$#@#$#@
NetLogo 5.0.5
@#$#@#$#@
@#$#@#$#@
@#$#@#$#@
@#$#@#$#@
@#$#@#$#@
default
0.0
-0.2 0 0.0 1.0
0.0 1 1.0 0.0
0.2 0 0.0 1.0
link direction
true
0
Line -7500403 true 150 150 90 180
Line -7500403 true 150 150 210 180

@#$#@#$#@
1
@#$#@#$#@
